```
(* gnuplot
  f(x)=a+b*x
  fit f(x)'Weinbau2.dat' using 3:2:($4) via a,b
  plot[78:128] f(x) ls 1 notitle,'Weinbau2.dat' using 3:2:4 with errorbar notitle *)
```

In[89]:=
```
SetDirectory[NotebookDirectory[]]
```

Out[89]=
```
/Users/distler/Desktop/teaching/Mathematica22
```

In[90]:=
```
data0 = Import["Weinbau2.dat"]
```

Out[90]=
```
{{1971, 5.6, 116.37, 0.4}, {1973, 3.2, 82.77, 0.2}, {1974, 4.5, 110.68, 0.3},
 {1975, 4.2, 97.5, 0.3}, {1976, 5.2, 115.88, 0.4}, {1977, 2.7, 80.19, 0.2},
 {1978, 4.8, 125.24, 0.3}, {1979, 4.9, 116.15, 0.3}, {1980, 4.7, 117.36, 0.3},
 {1981, 4.1, 93.31, 0.3}, {1982, 4.4, 107.46, 0.3}, {1983, 5.4, 122.3, 0.4}}
```

In[91]:=
```
data = {#[[3]], #[[2]]} & /@ data0
weights = 1 / #[[4]]^2 & /@ data0
```

Out[91]=
```
{{116.37, 5.6}, {82.77, 3.2}, {110.68, 4.5}, {97.5, 4.2},
 {115.88, 5.2}, {80.19, 2.7}, {125.24, 4.8}, {116.15, 4.9},
 {117.36, 4.7}, {93.31, 4.1}, {107.46, 4.4}, {122.3, 5.4}}
```

Out[92]=
```
{6.25, 25., 11.1111, 11.1111, 6.25, 25.,
 11.1111, 11.1111, 11.1111, 11.1111, 11.1111, 6.25}
```

In[93]:=
```
lmf = LinearModelFit[data, x, x, Weights → weights]
```

Out[93]=
```
FittedModel[ -1.20101 + 0.0523512 x ]
```

In[94]:=
```
lmf["Properties"]
```

Out[94]=
```
{AdjustedRSquared, AIC, AICc, ANOVATable, ANOVATableDegreesOfFreedom,
 ANOVATableEntries, ANOVATableFStatistics, ANOVATableMeanSquares,
 ANOVATablePValues, ANOVATableSumsOfSquares, BasisFunctions,
 BetaDifferences, BestFit, BestFitParameters, BIC, CatcherMatrix,
 CoefficientOfVariation, CookDistances, CorrelationMatrix,
 CovarianceMatrix, CovarianceRatios, Data, DesignMatrix,
 DurbinWatsonD, EigenstructureTable, EigenstructureTableEigenvalues,
 EigenstructureTableEntries, EigenstructureTableIndexes,
 EigenstructureTablePartitions, EstimatedVariance, FitDifferences,
 FitResiduals, Function, FVarianceRatios, HatDiagonal, MeanPredictionBands,
 MeanPredictionConfidenceIntervals, MeanPredictionConfidenceIntervalTable,
 MeanPredictionConfidenceIntervalTableEntries, MeanPredictionErrors,
 ParameterConfidenceIntervals, ParameterConfidenceIntervalTable,
 ParameterConfidenceIntervalTableEntries, ParameterConfidenceRegion,
 ParameterErrors, ParameterPValues, ParameterTable, ParameterTableEntries,
 ParameterTStatistics, PartialSumOfSquares, PredictedResponse,
 Properties, Response, RSquared, SequentialSumOfSquares,
 SingleDeletionVariances, SinglePredictionBands,
 SinglePredictionConfidenceIntervals, SinglePredictionConfidenceIntervalTable,
 SinglePredictionConfidenceIntervalTableEntries, SinglePredictionErrors,
 StandardizedResiduals, StudentizedResiduals, VarianceInflationFactors}
```

In[95]:=
```
lmf["ParameterTable"]
lmf["ANOVATable"]
```

Out[95]=

| | Estimate | Standard Error | t-Statistic | P-Value |
|---|---|---|---|---|
| 1 | −1.20101 | 0.601453 | −1.99684 | 0.0737751 |
| x | 0.0523512 | 0.00587046 | 8.91774 | $4.49385\times10^{-6}$ |

Out[96]=

| | DF | SS | MS | F-Statistic | P-Value |
|---|---|---|---|---|---|
| x | 1 | 107.375 | 107.375 | 79.526 | $4.49385\times10^{-6}$ |
| Error | 10 | 13.5018 | 1.35018 | | |
| Total | 11 | 120.876 | | | |

In[97]:=
```
A = {1, #[[3]]} & /@ data0
y = {#[[2]]} & /@ data0
W = DiagonalMatrix[weights] / 1.35018
```

Out[97]=
```
{{1, 116.37}, {1, 82.77}, {1, 110.68}, {1, 97.5}, {1, 115.88}, {1, 80.19},
 {1, 125.24}, {1, 116.15}, {1, 117.36}, {1, 93.31}, {1, 107.46}, {1, 122.3}}
```

Out[98]=
```
{{5.6}, {3.2}, {4.5}, {4.2}, {5.2}, {2.7}, {4.8}, {4.9}, {4.7}, {4.1}, {4.4}, {5.4}}
```

Out[99]=
```
{{4.62901, 0., 0., 0., 0., 0., 0., 0., 0., 0., 0., 0.},
 {0., 18.516, 0., 0., 0., 0., 0., 0., 0., 0., 0., 0.},
 {0., 0., 8.22936, 0., 0., 0., 0., 0., 0., 0., 0., 0.},
 {0., 0., 0., 8.22936, 0., 0., 0., 0., 0., 0., 0., 0.},
 {0., 0., 0., 0., 4.62901, 0., 0., 0., 0., 0., 0., 0.},
 {0., 0., 0., 0., 0., 18.516, 0., 0., 0., 0., 0., 0.},
 {0., 0., 0., 0., 0., 0., 8.22936, 0., 0., 0., 0., 0.},
 {0., 0., 0., 0., 0., 0., 0., 8.22936, 0., 0., 0., 0.},
 {0., 0., 0., 0., 0., 0., 0., 0., 8.22936, 0., 0., 0.},
 {0., 0., 0., 0., 0., 0., 0., 0., 0., 8.22936, 0., 0.},
 {0., 0., 0., 0., 0., 0., 0., 0., 0., 0., 8.22936, 0.},
 {0., 0., 0., 0., 0., 0., 0., 0., 0., 0., 0., 4.62901}}
```

In[100]:=
```
Va = Inverse[(Transpose[A].W.A)]
a = Va.Transpose[A].W.y
```

Out[100]=
```
{{0.361745, -0.00348554}, {-0.00348554, 0.0000344623}}
```

Out[101]=
```
{{-1.20101}, {0.0523512}}
```

In[102]:=
```
Print[a[[1, 1]], " +/- ", Sqrt[Va[[1, 1]]]]
Print[a[[2, 1]], " +/- ", Sqrt[Va[[2, 2]]]]
lmf["ParameterTable"]
```

```
-1.20101 +/- 0.601453
0.0523512 +/- 0.00587046
```

Out[104]=

| | Estimate | Standard Error | t-Statistic | P-Value |
|---|---|---|---|---|
| 1 | −1.20101 | 0.601453 | −1.99684 | 0.0737751 |
| x | 0.0523512 | 0.00587046 | 8.91774 | $4.49385\times10^{-6}$ |

In[105]:=
```
lmf2 = LinearModelFit[data, x, x,
  VarianceEstimatorFunction → (1 &), Weights → weights]
```

Out[105]=
```
FittedModel[ -1.20101 + 0.0523512 x ]
```

In[106]:=
```
lmf2["ParameterTable"]
```

Out[106]=

| | Estimate | Standard Error | t-Statistic | P-Value |
|---|---|---|---|---|
| 1 | −1.20101 | 0.517613 | −2.32028 | 0.0427525 |
| x | 0.0523512 | 0.00505215 | 10.3622 | $1.14617\times10^{-6}$ |

In[107]:=
```
lpt = ListPlot[{#[[3]], Around[#[[2]], #[[4]]]} & /@ data0];
plt = Plot[lmf2[x], {x, 80, 130}];
Show[lpt, plt]
```

Out[109]=